

Некоторые примеры типовых моделей процессов

Пример 10

Интервалы прихода клиентов в парикмахерскую с одним креслом распределены равномерно на интервале 18 ± 6 мин. Время стрижки также распределено равномерно на интервале 16 ± 4 мин. Клиенты приходят в парикмахерскую, стригутся в порядке очереди: «первым пришел – первым обслужился». Необходимо построить GPSS-модель парикмахерской, которая должна обеспечить сбор статистических данных об очереди. Промоделируйте работу парикмахерской в течение 8 часов.

Построение модели

Порядок блоков в модели соответствует порядку фаз, в которых клиент оказывается при движении в реальной системе:

- 1) клиент приходит;
- 2) если необходимо, ждет своей очереди;
- 3) садится в кресло парикмахера;
- 4) парикмахер обслуживает клиента;
- 5) клиент уходит из парикмахерской.

Единица модельного времени – 1 минута.

Программа:

```
** MODEL SEGMENT 1
GENERATE 18,6 ; Приход клиентов
QUEUE BARBERQ ; Присоединение к очереди
SEIZE BARBER ; Переход в кресло парикмахера
DEPART BARBERQ ; Выход из очереди
ADVANCE 16,4 ; Обслуживание у парикмахера
RELEASE BARBER ; Освобождение парикмахера
TERMINATE ; Уход из парикмахерской
** MODEL SEGMENT 2
GENERATE 480 ; Транзакт-таймер
TERMINATE 1 ; Завершение прогона

START 7
```

Пример 11

На фабрике в кладовой работает один кладовщик. Он выдает запасные части механикам, обслуживающим станки и устанавливающим эти части на сломанных станках. Ассортимент запчастей слишком велик для того, чтобы каждый механик мог иметь все запасные части в своем ящике. Время, необходимое для удовлетворения запроса кладовщиком, зависит от типа запасной части. Запросы бывают двух категорий:

Категория	Интервал времени между запросами, с	Время обслуживания поломки, с
1	420 ± 360	300 ± 90
2	360 ± 240	100 ± 30

Порядок обслуживания запросов кладовщиком такой: запросы первой категории обслуживаются только в том случае, если в очереди нет ни одного запроса второй категории. Внутри одной категории дисциплина обслуживания – «первым пришел – первым обслужился». Необходимо создать модель работы кладовой, моделирование выполнять в течение восьмичасового рабочего дня.

Построение модели

В задаче есть два различных типа заявок, поступающих на обслуживание к одному устройству. Различаются распределения интервалов приходов и времени обслуживания для этих типов заявок. Трудность заключается в том, что один из типов имеет преимущество в обслуживании. Построить модель можно с использованием средств задания заявкам различных приоритетов. Запросы первой категории будем моделировать одним сегментом, а запросы второй категории – другим. Разные относительные приоритеты задаются путем использования в операнде **E** блока **GENERATE** у запросов второй категории большего значения, чем у запросов первой категории.

Единица модельного времени – 1 секунда.

Программа:

```
** REQ OF TYPE 1
GENERATE 420,360,,,1 ;Приход запросов 1й категории
QUEUE QSTCKM1 ;Присоединение к очереди 1-й кат.
SEIZE STOCKMAN ;Занятие кладовщика
DEPART QSTCKM1 ;Выход из очереди
ADVANCE 300,90 ;Обслуживание кладовщиком
RELEASE STOCKMAN ;Освобождение кладовщика
TERMINATE ;Уход из кладовой
```

```
** REQ OF TYPE 2
GENERATE 360,240,,,2 ;Приход запросов 2й категории
QUEUE QSTCKM2
SEIZE STOCKMAN
DEPART QSTCKM2
ADVANCE 100,30
RELEASE STOCKMAN
TERMINATE
```

```
** Сегмент таймера
GENERATE 28800 ;Приход транзакта-таймера
TERMINATE 1
```

```
START 1 ;Запуск 1 прогона
```

Пример 12

Изготовление деталей определенного вида включает длительный процесс сборки, который заканчивается коротким периодом обжига в печи. Поскольку эксплуатация печи обходится очень дорого, несколько сборщиков используют одну печь, в которой одновременно можно обжигать только одну деталь. Сборщик не может начать новую сборку, пока не вытащит из печи предыдущую деталь.

Таким образом, сборщик работает в таком режиме:

- 1) собирает свою деталь;
- 2) ожидает возможности использования печи в очереди по принципу FIFO;
- 3) использует печь;
- 4) возвращается к п. 1.

Время, необходимое на выполнение различных операций, таково:

Сборка	30±5 мин
Обжиг	8±2 мин

Построение модели

В этой задаче есть два ограничивающих условия, влияющих на пропускную способность системы: одна печь и некоторое фиксированное количество работающих сборщиков.

Будем для моделирования печи использовать объект «устройство» (Facility), также отождествим сборщиков с транзактами. Тогда можно считать, что сборщики «циркулируют» в системе, периодически осуществляя сборку и обжиг. Аналогично транзакты должны циркулировать в GPSS-модели системы. Как видно из постановки задачи, модель представляет собой замкнутую СМО с одним устройством обслуживания.

В реальной системе, после того, как сборщик вынимает из печи обожженную деталь, он возвращается и начинает этап сборки очередной детали. В модели после того, как транзакт завершает использование устройства, он должен быть перемещен с помощью блока **TRANSFER** в блок следующей сборки. Для ограничения общего количества транзактов, циркулирующих в модели, необходимо использовать операнд **D** блока **GENERATE**.

Для вычисления прибыли при заданном числе сборщиков необходимо знать, сколько готовых деталей они сделают на протяжении моделируемого периода. Это значение равно количеству выходов из печи, наблюдаемое в атрибуте FC\$OVEN, т.е. в процессе моделирования нас интересует именно эта величина.

```
GENERATE ,,,4 ;Определение количества сборщиков
BACK1 ADVANCE 30,5 ;Сборка следующей детали
      SEIZE.OVEN ;Занятие печи
      ADVANCE 8,2 ;Использование печи
      RELEASE OVEN ;Освобождение печи
      TRANSFER ,BACK1 ; переход к сборке следующей детали
```

** Сегмент таймера

```
GENERATE 2400
TERMINATE 1
```

START 1 ;Запуск 1 прогона

Пример 13

Фабрика выпускает электронные часы. В упаковочном отделе часы упаковываются для отгрузки автоматической упаковочной машиной (роботом) в количестве, заказанном розничными продавцами.

Размер заказа определяется следующей зависимостью:

Частота	0.1	0.25	0.3	0.15	0.12	0.05	0.03
Размер заказа	6	12	18	24	30	36	48

Время между поступлением заказов распределено экспоненциально со средним значением 15 минут. Время упаковки заказа составляет 10 секунд на каждые часы, упакованные в заказе (положить в коробку), плюс 120 секунд на весь заказ (на коробку). Производство выпускает электронные часы партиями по 60 штук через каждые 45 ± 5 минут. Необходимо смоделировать 5 дней работы фабрики для получения следующих характеристик:

1. Среднее количество заказов, ожидающих выполнения в упаковочном отделе;
2. Распределение транзитного времени заказов.

Построение модели

В этой задаче есть два ограничивающих ресурса, влияющих на пропускную способность системы: один робот и склад часов. Будем для моделирования робота использовать объект «устройство» (Facility), а для моделирования склада используем объект «хранилище» (Storage).

** Единица времени равна одной минуте

```
Sizeorder FUNCTION RN1,D7 ; Размер заказа
      .1,6/.35,12/.65,18/.8,24/.92,30/.97,36/1.0,48
Transit TABLE M1,5,1,20 ; Транзитное время
Stock STORAGE 2000 ; Склад вмещает 2000 единиц
```

```

GENERATE (Exponential (1,0,15)) ; Поступление заказа
ASSIGN 1, FN$Sizeorder ; P1=размер заказа
TEST GE S$Stock, P1 ; Достаточно запасов?
LEAVE Stock, P1 ; забрать со склада
QUEUE Packing
SEIZE Machine ; Робот занимается упаковкой
DEPART Packing
ADVANCE (0.167#P1+2) ; Время упаковки
RELEASE Machine ; Робот освобождается
SAVEVALUE 1+, P1 ; Суммарное количество упакованных часов
TABULATE Transit ; Записывается время упаковки заказа
TERMINATE

```

```

GENERATE 45,5 ; Транзакт каждые 45+/-5 минут
ENTER Stock, 60 ; Произведено 60 часов, склад пополняется
TERMINATE

```

```

GENERATE ,, ,1,10 ; Транзакт для инициализации склада
ENTER Stock, 1000 ; Устанавливается начальный запас
TERMINATE

```

Пример 14

В СМО заявки поступают в одноканальное устройство с интервалом 100 единиц распределенным по экспоненциальному закону. Заявки обслуживаются в нем с интервалом 80 ± 30 . Пусть обслуживание 15% заявок оказывается неудачным, и заявки нуждаются в повторном обслуживании. Промоделировать обслуживание 5000 заявок.

Построение модели

```

TBL TABLE M1,0,150,50
GENERATE (exponential(1,0,100))
MM2 QUEUE 1
    SEIZE 1
    DEPART 1
    ADVANCE 80,30
    RELEASE 1
    TRANSFER .15, MM1, MM2
MM1 TABULATE TBL
    TERMINATE 1

START 5000

```

Пример 15

Моделируется железнодорожный переезд со шлагбаумом. Автомобили с каждой стороны переезда подъезжают со средним интервалом в 30 секунд, распределенным по экспоненциальному закону. Через шлагбаум одновременно могут переезжать два автомобиля. Время переезда 15 ± 3 секунды. В среднем через 1 час проходит поезд, и шлагбаум закрывается на 3 ± 1 минут. Промоделировать работу переезда в течение суток.

Построение модели

Блок GATE управляет потоком заявок на основе проверки значения условия. В GPSS имеются логические ключи, которые могут рассматриваться как ячейки, со значениями 0 или 1, или как элементы оборудования. Блоком присваивания для ключей является блок LOGIC. Состояние ключа может быть S (=1), R (=0) или I (инвертировать). С помощью команды INITIAL атрибуту объекта можно сопоставить начальное значение, отличное от 0.

```
INITIAL LS$SHLAG,1
PRZHD STORAGE 2
TT1 TABLE M1,0,10,50
TT2 TABLE (Q1+Q2),0,3,50

GENERATE (exponential (1,0,30))
QUEUE 1
GATE LS SHLAG
ENTER PRZHD
DEPART 1
ADVANCE 15,3
LEAVE PRZHD
TABULATE TT1
TERMINATE

GENERATE (exponential (1,0,30))
QUEUE 2
GATE LS SHLAG
ENTER PRZHD
DEPART 2
ADVANCE 15,3
LEAVE PRZHD
TABULATE TT1
TERMINATE

GENERATE 3600,900
LOGIC R SHLAG
ADVANCE 180,60
LOGIC S SHLAG
TABULATE TT2
TERMINATE

GENERATE (60#60#24)
TERMINATE 1

START 1
```

Пример 16

Пусть сообщения поступают в систему со средним интервалом времени в 750 единиц. Время распределено по экспоненциальному закону. Пусть число пакетов, на которые разбивается сообщение, равномерно распределено в интервале от 3 до 12. Пусть на передачу одного пакета по линии связи требуется 90 ± 20 единиц времени. После передачи каждого пакета по линии связи, все они последовательно обрабатываются устройством, объединяясь в одно сообщение. Обработка в устройстве занимает 80 ± 50 единиц времени. Провести моделирование передачи и обработки 1000 сообщений.

Построение модели

Для организации циклов можно использовать блок LOOP, который имеет формат LOOP A,B

Здесь A – номер или имя P-параметра, используемого для организации цикла, а B – метка блока, куда будет послана заявка, если цикл еще не завершен. Обычно метка расположена в модели выше блока LOOP. Когда транзакт входит в блок LOOP, то значение P-параметра уменьшается на 1, и если оно все еще больше 0, то транзакт переходит по метке. Если оно равно 0, то транзакт проходит сквозь блок и следует дальше по модели.

```
GENERATE (EXPONENTIAL(1,0,750))
  ASSIGN 1,(DUNIFORM(1,3,12))
  ASSIGN 2,P1
  QUEUE OCH
  SEIZE LINE1
  DEPART OCH
  ADVANCE 90,20
  TRANSFER ,BBE2
BBC2 SPLIT 1,BG2,3
  ADVANCE 90,20
BBE2 LOOP 1,BBC2
  RELEASE LINE1
BG2 QUEUE UST
  SEIZE UST
  DEPART UST
  ADVANCE 80,50
  RELEASE UST
  ASSEMBLE P2
  TERMINATE 1

START 1000
```

Пример 17

Пусть в СМО имеется 5 обслуживающих устройств. Нужно отправлять заявку в любое устройство, у которого длина очереди меньше 2. Если требуемого устройства нет, то заявки должны удаляться из системы. Заявки приходят с интервалом 10 ± 8 , а обслуживание занимает 49 ± 40 единиц времени. Промоделировать обслуживание 1000 заявок.

Построение модели

```
GENERATE 10,8
  SELECT L 1,1,5,2,Q,MET1
  QUEUE P1
  SEIZE P1
  DEPART P1
  ADVANCE 49,40
  RELEASE P1
  TERMINATE 1
MET1 TERMINATE

START 1000
```

Пример 18

В СМО поступают заявки со средним интервалом 100 единиц времени, распределенным по экспоненциальному закону. Время обслуживания заявок составляет 48 ± 20 единиц времени. Устройство имеет интервалы доступности и недоступности, распределенные по нормальному закону со средним значением 2000 и отклонением в 200 единиц. Промоделировать работу системы за 20 циклов доступности и недоступности.

Построение модели

Блок FUNAVAIL предназначен для перевода одноканального устройства в недоступное для работы состояние, т.е. недоступно для блока SEIZE. Блок FAVAIL предназначен для возврата одноканального устройства в доступное для работы состояние.

```
GENERATE (EXPONENTIAL (1,0,100))
QUEUE UST
SEIZE UST
DEPART UST
ADVANCE 48,20
RELEASE UST
TERMINATE
**
GENERATE ,,,1
RET FAVAIL UST
ADVANCE (NORMAL(1,2000,200))
FUNAVAIL UST
ADVANCE (NORMAL(1,2000,200))
SPLIT 1,RET
TERMINATE 1

START 20
```

Пример 19

В этой задаче моделируется работа сверлильного станка с периодической заменой запчасти. На сверлильном станке сверло периодически выходит из строя (затупляется). Токарь-станочник меняет его на запасное, если оно есть, и передает слесарю-инструментальщику для восстановления (заточки). Слесарь в основном занят изготовлением техоснастки и выполняет заточку сверла в свободное от основной работы время.

Необходимо промоделировать работу станка в течение года. Найти загрузку слесаря и станка, количество замен сверла, построить график ожидания сверлом операции по заточке.

Построение модели

```
** модельный сегмент 1
INITIAL 1,1 ;есть в запасе сверло
GENERATE ,,,1 ;токарь появился
Again SEIZE Mac ;станок включили
ADVANCE (Normal(1,3500,700)) ;ждем поломки сверла
```

```

RELEASE Mac ;станок выключили
ADVANCE 40 ;извлекли сверло
SPLIT 1,Fetch ;отправили сверло слесарю
QUEUE Fix
SEIZE Fixer ;слесарь начал восстановление сверла
DEPART Fix
ADVANCE (Normal(1,80,5)) ;длительность восстановления
RELEASE Fixer ;слесарь освободился
SAVEVALUE 1+,1 ;есть сверло в запасе
TERMINATE ;процесс заточки завершен
Fetch TEST G X1,0 ;проверяем есть ли сверло в запасе
SAVEVALUE 1-,1 ;в запасе на одно сверло меньше
ADVANCE 60 ;устанавливаем сверло
TRANSFER ,Again ;можно снова включать станок
** модельный сегмент 2
GENERATE (Exponential(1,0,90)),,,1 ;поток основной работы
SEIZE Fixer ;слесарь занят делом
ADVANCE 80,40 ;время работы
RELEASE Fixer ; слесарь освободился
TERMINATE ;
** модельный сегмент 3
GENERATE 104000 ;сегмент-таймер
TERMINATE 1

START 1

```